

Comparison of PROC MIXED and PROC GLM for Analysis of Repeated Measures Data

Hossein N. Yarandi, PhD, University of Florida, Gainesville, Florida

ABSTRACT

Repeated measures design utilizes the same subject (person, animal, plant, store, test market, etc.) for each of the treatments under study. Thus, the subject serves as a block, and the experimental units within a block may be viewed as the different occasions when a treatment is applied to the subject. A repeated measures study may involve several treatments or only a single treatment that is evaluated at different points in time. Traditionally, PROC GLM in SAS is utilized to address univariate and multivariate approaches of repeated measures analyses of variance. PROC GLM was originally written as a fixed-effects procedure that would handle models with both classification and regression variables. On the other hand, PROC MIXED procedure employs a more general covariance structure approach. In this paper, similarities and differences between the two approaches will be discussed through a numerical example.

INTRODUCTION

A repeated measures study may involve several treatments or only a single treatment that is evaluated at different points in time. Subjects used in repeated measures studies in the behavioral and life sciences include persons, households, observers, and experimental animals. At other times the subjects in repeated measures designs are store, test markets, cities, and plants. The study units that are used in repeated measures are usually referred to as subjects. When several measurements are taken on the same subject (or experimental unit), the measurements tend to be correlated with each other. To account for the correlation of the multiple measures on experimental units, multivariate analysis of variance should be utilized.

If questions about changes across the various measurements taken on an experimental unit are important, then we are dealing with repeated measures data. What distinguishes repeated measures data from any other multivariate data is not so much the existence of the repeated measurements but the desire to examine changes in the measurements taken on each subject. These changes are therefore known as within-subject effects. In repeated measures analysis of variance, the effects of interest are: between-subject effects, within-subject effects, and within-subject-by-between-subject interactions. For tests that involve only between-subject effects, both the multivariate and univariate approaches give rise to same tests. For within-subject effects and for within-subject-by-between-subject interaction effects,

the univariate and multivariate approaches may yield different tests (Kirk, 1982).

Both PROC GLM and PROC MIXED test within subject variability for repeated measures analysis of variance. PROC GLM is basically a fixed-effects procedure that can handle class and continuous variables. With the statements such as TEST, RANDOM, and REPEATED, PROC GLM can be used to test mixed and repeated measures applications. PROC GLM can provide results of the multivariate and univariate repeated measures analyses and multivariate and univariate analyses of contrasts. On the other hand, PROC MIXED applies methods based on the mixed model with special parametric structure on the covariance matrices. A mixed model is a generalization of the standard model used in the GLM procedure, the generalization being that the data are permitted to exhibit correlation and variability. It has the flexibility of modeling the variances and covariances as well as the means of a data set. Although both PROC GLM and PROC MIXED contain the same statements such as RANDOM and REPEATED, the functions of the statements differ greatly between the two procedures that result in different output.

EXAMPLE

Ten subjects agreed to participate in a study to examine the concentration of drug in the bloodstream for two different dosage forms (tablet and capsule) of the same product following a single dose. Five subjects were allocated at random to the capsule form and the other five to the tablet form. Blood sample were obtained immediately preceding the assigned dose and at 1, 2, 3, 4, and 5 hours after dosing, and were analyzed for the concentration of the drug product in the bloodstream. The SAS Codes and the data (in mg/ml) are shown below.

```
DATA GLM;
INPUT SUBJECT DRUG $ HOUR1-HOUR5;
DATALINES;
1 TABLET 50 75 120 60 30
2 TABLET 40 80 135 70 40
3 TABLET 55 75 125 85 50
4 TABLET 70 85 140 90 40
5 TABLET 60 90 150 95 50
6 CAPSULE 30 55 80 130 65
7 CAPSULE 25 50 75 125 60
8 CAPSULE 35 65 85 140 85
9 CAPSULE 45 70 90 145 80
10 CAPSULE 50 75 95 160 90
;
```

The following SAS codes provide the multivariate and univariate repeated measures ANOVA using PROC GLM.

```
PROC GLM;
  CLASS DRUG;
  MODEL HOUR1-HOUR5 = DRUG
    / NOUNI;
  REPEATED HOUR / PRINTE;
  REPEATED HOUR 5 CONTRAST(1)
    / PRINTM SUMMARY NOU NOM;
  REPEATED HOUR 5 PROFILE
    / PRINTM SUMMARY NOU NOM;
RUN;
```

The class (or independent) variable is DRUG and is between-subject fixed effect on the MODEL statement. All five repeated measures response variables (HOUR1-HOUR5) are placed on the left hand side of the MODEL statement. The NOUNI option suppresses the printing of one factor analysis of variance for each of the five response variables. The REPEATED statement tests the within-subjects repeated measures effect of time and provides the test of interaction between drug and time. The name HOUR in the REPEATED statement associates a name with the factor defined by the variability across the dependent variables for an observation. In this example, the concentration of the drug product in the bloodstream are measured at five consecutive hours, so the factor defined across the variables is named HOUR. PRINTE option on the REPEATED statement provides the correlation and partial correlations matrices for HOUR1-HOUR5 and a test for sphericity. There are several transformations for generating contrast variables that have proven to be extremely useful in repeated measures analysis of variance which can be generated automatically by the REPEATED statement in PROC GLM. The univariate and multivariate statistics and probabilities are not affected by the choice of transformation; only the elements of the matrices involved in the tests and the ANOVA tables produced by the SUMMARY option change when a different transformation is used. The two transformations that are used here are CONTRAST and PROFILE. CONTRAST transformation is useful when one level of the repeated measures is to be compared against the rest and the PROFILE transformation compares the successive levels of a repeated measure.

PROC GLM prints the levels of class variable (DRUG), total number of observations, the number of repeated measurements, and partial correlations of the original (untransformed) dependent variables (not shown here). These correlations are corrected for the independent variable (DRUG) in the model. It shows the Error SS&CP Matrix, which are the error sums of squares matrix for the transformed variables. In this case, the default transformation subtracts the measurement

representing the last hour (HOUR5) from each of the other hours. The main diagonal of the Error SS&CP Matrix are sums of squares and the off diagonal of the matrix are the cross products of the transferred variables. In addition, using the PRINTE option, SAS output shows the partial correlation coefficients of the transformed variables.

One of the assumptions required for any univariate F ratio to be distributed as the central (tabled) F is that of compound symmetry of the variance-covariance matrix. When the treatments are independent the covariances are always zero. With repeated measures designs, however, the covariances will not be zero. However, a pattern of constant variances and covariances is referred to as compound symmetry. The relationship between the variances and covariances is irrelevant. The assumption of compound symmetry of the variance-covariance matrix represents a sufficient condition underlying the analysis of variance. On the other hand, if the standard errors of the differences between pairs of time interval means are constant, then sphericity exists. Sphericity is met automatically if the variance-covariance matrix exhibits compound symmetry. Two sphericity tests are presented in Output 1; the first test uses the particular transformation and the second uses a set of orthogonal components. The test applied to the orthogonal components is the one that is important in determining whether the univariate F tests for the within-subjects effects are valid. When there are only two levels of a within-subjects effect, there is only one transformed variable, and a sphericity test is not needed and cannot be performed. The null and alternative hypotheses for the above tests are: sphericity does not exist versus sphericity exists. If the null hypothesis is not rejected, the univariate repeated measures analysis of variance would also be appropriate. If the null hypothesis is rejected within a 0.05 and a 0.0001 p-value, still the univariate repeated measures analysis of variance could be applied with an adjustment. If the null hypothesis of sphericity is rejected so dramatically; i.e., p-value < 0.0001, the multivariate repeated measures analysis of variance must only be applied. In this example, the 'Applied to Orthogonal Components' test results in $\chi^2 = 13.757$, $p = 0.1312$ (Output 1), therefore, we conclude that the sphericity exists, i.e., the results of the multivariate repeated measures analysis in this example are not drastically different from the ones using the univariate repeated measures analysis. The univariate approach, if appropriate, is more powerful than its multivariate counterpart (Littell, Stoup, & Freund, 2002).

As shown in Output 2, we conclude that there is a significant change in mean drug concentration in the bloodstream over the five hours ($F = 211.55$, $p < 0.0001$). The test for HOUR*DRUG tells us if the effect of within-subjects changes in drug concentration in the bloodstream is different from the between-subjects changes. The result indicates that the pattern of change in drug

concentration in the bloodstream between tablet and capsule are significantly different over the five hours ($F = 180.31, p < 0.0001$).

One additional test provided by the repeated statement is labeled tests of hypotheses for between subjects effects (output 3), and it tests the hypothesis that these effects (in this example, difference in mean drug concentration in the bloodstream between the tablet and capsule) have no effect on the dependent variables (drug concentration in each hour), ignoring the within-subjects effects in the design. Since these tests ignore the effects that exist across the dependent variables, they are constructed by simply adding together the dependent variables and performing an analysis on the sum. Actually, the analysis is performed on the sum divided by the square root of the number of dependent variables. Thus, the between-subjects tests differ from the corresponding multivariate tests of within-subjects in that they do not attempt to account for correlation among the dependent variables. They only tell us whether these factors (e.g., drug) are important when averaged over the dependent variables representing the within-subjects effects. In this example, we conclude that there is no significant difference in mean drug concentration in the bloodstream between the tablet and capsule ($F = 0.08, p = 0.7810$). That is, the effect of tablet and capsule is not significantly different when the drug concentration in the bloodstream is averaged over the five hours. There is no difference in the results of the test of hypothesis for between-subjects effects using either univariate or multivariate analyses.

PROC GLM and REPEATED statement provide both the multivariate (Output 2) and univariate analyses (Output 4). Since the Huynh-Feldt ($H\epsilon$) Epsilon is greater than one (1.3610), then no adjustment of the p-values of the univariate F's is needed. This was expected, since the p-value of the test for sphericity was higher than 0.05 ($p = 0.1312$, Output 1).

PROC MIXED

Unlike PROC GLM, by using PROC MIXED, we can omit between-within interaction effects and can use continuous variables in within-subject effects. PROC MIXED allows us to specify and fit a reduced model. The covariance structure in PROC MIXED should be specified in advance. Three of the most commonly used covariance structures are compound symmetric, autoregressive order one, and unstructured. We can specify these covariance structure as CS, AR(1), and UN, respectively, in the REPEATED statement of PROC MIXED. In order to use PROC MIXED, we must transform the data set from a multivariate mode to a univariate mode (Littell, Milliken, Stroup & Wolfinger, 1996; Khuri, Mathew, & Sinha, 1998). A univariate form of the data is shown below in a SAS data set named

MIXED that contains variables SUBJECT, DRUG, HOUR, DC (Drug Concentration). The objectives are to compare the drug concentration between tablet and capsule over time.

```
DATA MIXED; SET GLM;
  HOUR = 1; DC = HOUR1; OUTPUT;
  HOUR = 2; DC = HOUR2; OUTPUT;
  HOUR = 3; DC = HOUR3; OUTPUT;
  HOUR = 4; DC = HOUR4; OUTPUT;
  HOUR = 5; DC = HOUR5; OUTPUT;
PROC MIXED;
  CLASS DRUG HOUR SUBJECT;
  MODEL DC = DRUG HOUR DRUG*HOUR;
  REPEATED / TYPE = CS SUB = SUBJECT;
PROC MIXED;
  CLASS DRUG HOUR SUBJECT;
  MODEL DC = DRUG HOUR DRUG*HOUR;
  REPEATED / TYPE = AR(1) SUB =
SUBJECT;
PROC MIXED;
  CLASS DRUG HOUR SUBJECT;
  MODEL DC = DRUG HOUR DRUG*HOUR;
  REPEATED / TYPE = UN SUB = SUBJECT;
RUN;
```

All three mixed procedures have the same class level information (Output 5), however the results of the tests of fixed effects are different from one another. These results are shown in Output 6, 7, and 8.

The difference in F and p values in Output 6, 7, and 8 is due to the way that PROC MIXED computes the F statistics under different covariance structure. Two model-fit criteria computed by PROC MIXED, Akaike's Information Criterion (AIC) and Schwarz' Bayesian Criterion (SBC) can be used to determine which of the three covariance structure models to choose for final inference. The following table shows the AIC and SBC values for the three covariance structures provided by the SAS output (not shown here).

The AIC and SBC values are fairly close for the three covariance structures, however, the two values for UN structure is marginally smaller (or closet to zero). On this basis, we can choose UN and report the F and p-values of fixed effect models (Output 8).

| | Covariance Structure | | |
|-----|----------------------|-------|-------|
| | CS | AR(1) | UN |
| AIC | 292.7 | 292.1 | 288.2 |
| SBC | 293.3 | 292.7 | 292.6 |

PROC MIXED does not exactly duplicate any of the PROC GLM transformations for generating contrast variables. However, PROC MIXED can produce more complete results along with valid tests with properly

specified covariance structure. Finally, missing data can be handled more effectively with PROC MIXED as long as the missing data are random (Littell, Milliken, Stroup, & Wolfinger, 1996).

REFERENCES

Khuri, A., Mathew, T., & Sinha, B. (1998). Statistical Tests for Mixed Linear Models. New York: John Wiley & Sons, Inc.

Kirk, R. (1982). Experimental Design, Pacific Grove, CA: Brooks/Cole Publishing

Littell, R., Milliken, G., Stroup, W., & Wolfinger, R.

(1996). SAS System for Mixed Models, NC: SAS Institute Inc.

Little, R., Stroup, W., Freund, R. (2002). SAS for Linear Models, NC: SAS Institute Inc.

SAS Institute Inc. (2001). SAS/STAT User's Guide, Cary NC: SAS Institute Inc.

Output 1. PROC GLM Sphericity Tests

| Sphericity Tests | | | | |
|-----------------------|----|-----------|----------------------|------------|
| Variables | DF | Criterion | Mauchly's Chi-Square | Pr > ChiSq |
| Transformed Variates | 9 | 0.0252939 | 23.59531 | 0.0050 |
| Orthogonal Components | 9 | 0.1171914 | 13.75699 | 0.1312 |

Output 2. PROC GLM Multivariate Tests

| Manova Test Criteria and Exact F Statistics for the Hypothesis of no HOUR Effect H = Type III SSCP Matrix for HOUR E = Error SSCP Matrix | | | | | |
|---|--------------|---------|--------|--------|--------|
| Statistic | Value | F Value | Num DF | Den DF | Pr > F |
| Wilks' Lambda | 0.00587399 | 211.55 | 4 | 5 | <.0001 |
| Pillai's Trace | 0.99412601 | 211.55 | 4 | 5 | <.0001 |
| Hotelling-Lawley Trace | 169.24212349 | 211.55 | 4 | 5 | <.0001 |
| Roy's Greatest Root | 169.24212349 | 211.55 | 4 | 5 | <.0001 |
| Manova Test Criteria and Exact F Statistics for the Hypothesis of no HOUR*DRUG Effect H = Type III SSCP Matrix for HOUR*DRUG E = Error SSCP Matrix | | | | | |
| Statistic | Value | F Value | Num DF | Den DF | Pr > F |
| Wilks' Lambda | 0.00688490 | 180.31 | 4 | 5 | <.0001 |
| Pillai's Trace | 0.99311510 | 180.31 | 4 | 5 | <.0001 |
| Hotelling-Lawley Trace | 144.24534293 | 180.31 | 4 | 5 | <.0001 |
| Roy's Greatest Root | 144.24534293 | 180.31 | 4 | 5 | <.0001 |

Output 3. PROC GLM Between Subject Tests

| The GLM Procedure Repeated Measures Analysis of Variance Tests of Hypotheses for Between Subjects Effects | | | | | |
|---|----|-------------|-------------|---------|--------|
| Source | DF | Type III SS | Mean Square | F Value | Pr > F |
| DRUG | 1 | 40.500000 | 40.500000 | 0.08 | 0.7810 |

| | | | |
|-------|---|-------------|------------|
| Error | 8 | 3920.000000 | 490.000000 |
|-------|---|-------------|------------|

Output 4. PROC GLM Univariate Tests

| The GLM Procedure | | | | | | |
|---|--------------|----------------------------|-------------|---------|------------|--|
| Repeated Measures Analysis of Variance | | | | | | |
| Univariate Tests of Hypotheses for Within Subject Effects | | | | | | |
| Source | DF | Type III SS | Mean Square | F Value | Pr > F | |
| HOUR | 4 | 34288.00000 | 8572.00000 | 279.90 | <.0001 | |
| HOUR*DRUG | 4 | 19472.00000 | 4868.00000 | 158.96 | <.0001 | |
| Error (HOUR) | 32 | 980.00000 | 30.62500 | | | |
| | | | | | Adj Pr > F | |
| | Source | | G - G | | H - F | |
| | HOUR | | <.0001 | | <.0001 | |
| | HOUR*DRUG | | <.0001 | | <.0001 | |
| | Error (HOUR) | | | | | |
| | | Greenhouse-Geisser Epsilon | | 0.7374 | | |
| | | Huynh-Feldt Epsilon | | 1.3610 | | |

Output 5. PROC MIXED Class Level Information

| Class Level Information | | |
|-------------------------|--------|----------------------|
| Class | Levels | Values |
| DRUG | 2 | CAPSULE TABLET |
| HOUR | 5 | 1 2 3 4 5 |
| SUB | 10 | 1 2 3 4 5 6 7 8 9 10 |

Output 6. PROC MIXED Tests of Fixed Effects Using CS

| Type 3 Tests of Fixed Effects | | | | |
|-------------------------------|--------|--------|---------|--------|
| Effect | Num DF | Den DF | F Value | Pr > F |
| DRUG | 1 | 8 | 0.08 | 0.7810 |
| HOUR | 4 | 32 | 279.90 | <.0001 |
| DRUG*HOUR | 4 | 32 | 158.96 | <.0001 |

Output 7. PROC MIXED Tests of Fixed Effects Using AR(1)

| Type 3 Tests of Fixed Effects | | | | |
|-------------------------------|-----------|-----------|---------|--------|
| Effect | Num DF | Den DF | F Value | Pr > F |
| DRUG | 1 | 8 | 0.09 | 0.7689 |
| HOUR | 4 | 32 | 233.95 | <.0001 |
| DRUG*HOUR | 4 | 32 | 166.61 | <.0001 |

Output 8. PROC MIXED Tests of Fixed Effects Using UN

| Type 3 Tests of Fixed Effects | | | | |
|-------------------------------|-----------|-----------|---------|--------|
| Effect | Num DF | Den DF | F Value | Pr > F |
| DRUG | 1 | 8 | 0.08 | 0.7810 |
| HOUR | 4 | 8 | 338.48 | <.0001 |
| DRUG*HOUR | 4 | 8 | 288.49 | <.0001 |

CONTACT INFORMATION

Hossein N. Yarandi, PhD
Associate Professor
University of Florida
Campus Box 100187
Gainesville, FL 32610-0187

Voice: (352) 273 - 6354
Fax: (352) 273 - 6505
E-mail: yarandi@ufl.edu